
POST TEST II

___________________Transformasi Geometri 

Tujuan

• Memberikan pengertian perlunya dilakukan transformasi gambar

• Memberikan contoh transformasi gambar

Teori Penunjang

Gambar Percobaan

Prosedur Percobaan
1. Buat Project dengan nama Project1 yang berisi Form1. Sediakan/tambahkan pada 

Project1 file module dengan nama Module1.

2. Tambahkan pada Form1 beberapa komponen,

a. DriveListBox dengan nama Drive1

b. DirListBox dengan nama Dir1

c. FileListBox dengan nama File1

d. PictureBox dengan nama Picture1

e. TextBox dengan nama Text1

f. CommandButton dengan nama Command1. Ubah Caption dari Command1 

dengan ”Geser Y”

3. Masukkan program berikut pada Module1.
Option Explicit

Public Declare Function SetPixel Lib "gdi32" ( _
    ByVal hdc As Long, ByVal x As Long, ByVal y As Long, _
    ByVal crColor As Long) As Long
Public Declare Function GetPixel Lib "gdi32" ( _
    ByVal hdc As Long, ByVal x As Long, ByVal y As Long) As Long

Global vImage() As Long
4. Masukkan Program berikut pada Form1.

Option Explicit

Private Sub Command1_Click()
    Command1.Enabled = False
    GeserY
    Command1.Enabled = True
End Sub

Private Sub Dir1_Change()
    File1.Path = Dir1.Path
End Sub

Private Sub Drive1_Change()
    Dir1.Path = Drive1.Drive
End Sub

Private Sub File1_Click()
    Picture1.Picture = LoadPicture(File1.Path + "\" + File1.FileName)
    ReDim vImage(0 To Picture1.ScaleWidth - 1, 0 To Picture1.ScaleHeight - 1) 
As Long
End Sub

Private Sub Form_Load()
    File1.Pattern = "*.bmp;*.jpg;*.jpeg;*.gif;*.tif"
    Picture1.ScaleMode = 3
    Picture1.AutoSize = False
    Picture1.AutoRedraw = True
End Sub

Sub GeserY()
    Dim Geser As Integer, x As Integer, y As Integer
    Geser = Val(Text1)
    Text1 = Geser

    For y = 0 To Picture1.ScaleHeight - 1
        For x = 0 To Picture1.ScaleWidth - 1
            vImage(x, y) = 0
        Next
    Next
    For y = 0 To Picture1.ScaleHeight - 1
        For x = 0 To Picture1.ScaleWidth - 1
            If y + Geser >= 0 And y + Geser < Picture1.ScaleHeight Then
                vImage(x, y + Geser) = GetPixel(Picture1.hdc, x, y)
            End If
        Next
    Next
    For y = 0 To Picture1.ScaleHeight - 1
        For x = 0 To Picture1.ScaleWidth - 1
            SetPixel Picture1.hdc, x, y, vImage(x, y)
        Next
    Next
    Picture1.Refresh
End Sub

5. Untuk mempercepat proses eksekusi, jadikan program kedalam bentuk file .exe.

6. Jalankan dari file .exe-nya.

7. Dari dalam program, ambil/buka sebuah gambar dengan ukuran yang tidak terlalu 

besar (misalkan 320 x 240).

8. Isi nilai pergeseran dengan nilai tertentu, misalkan 10. Tekan tombol geser untuk 

memulai pergeseran.

